


ME201/MTH281/ME400/CHE400

Lord Kelvin and the Age of the Earth



Lord Kelvin (1824 - 1907)

1. About Lord Kelvin

Lord Kelvin was born William Thomson in Belfast Ireland in 1824. He attended Glasgow University from the age of 10, and later took his BA at Cambridge. He was appointed Professor of Natural Philosophy at Glasgow in 1846, a position he retained the rest of his life. He worked on a broad range of topics in physics, including thermodynamics, electricity and magnetism, hydrodynamics, atomic physics, and earth science. He also had a strong interest in practical problems, and in 1866 he was knighted for his work on the transatlantic cable. In 1892 he became Baron Kelvin, and this name survives as the name of the absolute temperature scale which he proposed in 1848.

During his long career, Kelvin published more than 600 papers. He was elected to the Royal Society in 1851, and served as president of that organization from 1890 to 1895.

The information in this section and the picture above were taken from a very useful web site called the MacTutor History of Mathematics Archive, sponsored by St. Andrews University. The web address is

<http://www-history.mcs.st-and.ac.uk/~history/>

2. The Age of the Earth

The earth shows its age in many ways. Some techniques for estimating this age require us to observe the present state of a time-dependent process, and from that observation infer the time at which the process started. If we believe that the process started when the earth was formed, we get an estimate of the earth's age. Kelvin's method is of this type, and it focuses on the thermal state of the earth, which we discuss next.

The surface of the earth is a receptor for solar energy. The total solar flux incident on the earth, divided by the total area of the earth, is about $1.4 \times 10^3 \text{ W/m}^2$. This incident flux is partially reflected and partially absorbed. The part absorbed is in turn re-radiated. The energy balance is further complicated by the conductive and convective exchange between the earth and the atmosphere, and by the absorption of some of the re-radiated flux by the atmosphere. The net effect of all this is an average surface temperature T_s , which is approximately $0 \text{ }^\circ\text{C}$. We define this value for *Mathematica*, for use in our calculations below. Because we wish to preserve the dependence of our results on the parameter symbols, we use one symbol for the name of a parameter, and a modified symbol for the value of that parameter. In each case, the value is assigned to the name augmented by "val." Thus T_{sval} is the value of the parameter T_s .

```
In[1]:= Tsval = 0; (** °C **)
```

Kelvin assumed, as had others before him, that the earth originated as a molten sphere. We take the initial temperature T_i of this sphere to be $2000 \text{ }^\circ\text{C}$:

```
In[2]:= Tival = 2000; (** °C **)
```

The surface of such a molten globe will cool very rapidly. In a geologically negligible time, the radiative balance with the sun, as described above, will bring the surface temperature to the value T_s . Because the interior remains hot, there will be a conductive flux to the surface. This flux is far smaller than the solar flux. Present-day measurements give an average value (with considerable heterogeneity) of about $6.7 \times 10^{-2} \text{ W/m}^2$. Although this flux is energetically negligible in the surface energy balance, a knowledge of it can be the basis for an estimate of the earth's age -- a point noted by several authors before Kelvin, including Fourier and Buffon. Kelvin was the first to treat this idea with any precision. The idea is elegantly simple. The temperature gradient at the earth's surface associated with this flux will decrease with time, and can be determined at the present time by temperature measurements in mines. We can then use theory to calculate the time-dependent gradient, and then determine the time at which the theoretical gradient matches the present-day observed gradient. That time is then the age of the earth. We denote the measured value of this gradient by α . Its present value (again with considerable heterogeneity) is

```
In[3]:= aval = 0.037; (** °C/m **)
```

The final important parameter in the problem is the thermal diffusivity D_f of the material in the earth's crust. Once again there is heterogeneity, but a typical value is

```
In[4]:= Dfval = 1.2 * 10^-6; (** m^2/s **)
```

At several points in our calculations below, we need the number of seconds in a year:

```
In[5]:= speryr = 365 * 24 * 60 * 60
```

```
Out[5]= 31 536 000
```

3. The Flat Earth Approximation

In principle, we should solve the cooling problem for a sphere, with prescribed initial temperature T_i and prescribed surface temperature T_s . In practice we can solve a simpler problem. We know from previous work with the heat equation that a cooling process initiated at a boundary will in time t penetrate a distance into the material of the order of $\sqrt{D_f t}$. If we take for t the longest of the times that are suggested for the age of the earth -- namely 5×10^9 years -- we get a penetration depth (converted to km) of

$$\text{In[6]:= } \sqrt{\text{Dfval} * (5 * 10^9) * \text{speryr}} / 1000$$

$$\text{Out[6]= } 434.99$$

This number is very much smaller than the radius of the earth (about 6400 km). We conclude that the cooling process is confined to the surface layers, even for these very long times. Thus we may replace the spherical earth with a flat earth. In that case, the cooling problem is specified by the following boundary value problem for temperature T as a function of time t and distance x beneath the surface of the earth:

$$\frac{\partial T}{\partial t} = Df \frac{\partial^2 T}{\partial x^2}, \quad 0 < x < \infty, \quad t > 0,$$

$$\text{with } T(x, 0) = Ti, \text{ and } T(0, t) = Ts.$$

4. Solution of the Problem

This is a standard problem in one-dimensional transient heat conduction, and it may be solved by a Laplace transform or by similarity methods. In either case, the solution is obtained in terms of an error function and is given by

$$\text{In[7]:= } T[\mathbf{x_}, \mathbf{t_}] := Ts + (Ti - Ts) * \text{Erf}\left[\frac{\mathbf{x}}{2 * \sqrt{Df * t}}\right]$$

From this solution, we may calculate $\beta[t]$, the time-dependent temperature gradient at the surface $x = 0$:

$$\text{In[8]:= } \beta[\mathbf{t_}] := D[T[\mathbf{x}, \mathbf{t}], \mathbf{x}] /. \mathbf{x} \rightarrow 0$$

$$\text{In[9]:= } \beta[\mathbf{t}]$$

$$\text{Out[9]= } \frac{Ti - Ts}{\sqrt{\pi} \sqrt{Df t}}$$

This gradient is then equal to the present-day measured value α when t is equal to the age of the earth, which we call tage . By setting β equal to α and solving for t , we generate one of the most dramatic formulas of 19th century physics -- a formula for the age of the earth:

$$\text{In[10]:= } \text{tage} = \text{Simplify}[\mathbf{t} /. \text{Flatten}[\text{Solve}[\beta[\mathbf{t}] == \alpha, \mathbf{t}]]]$$

$$\text{Out[10]= } \frac{(Ti - Ts)^2}{Df \pi \alpha^2}$$

Now the moment of truth for Kelvin's theory: we calculate this age, in years, for our previous choice of parameter values.

$$\text{In[11]:= } \text{tageval} = (\text{tage} /. \text{Thread}[\{Ti, Ts, Df, \alpha\} \rightarrow \{\text{Tival}, \text{Tsva}, \text{Dfval}, \alpha\al}]) / (\text{speryr})$$

$$\text{Out[11]= } 2.45764 \times 10^7$$

We see that Kelvin's theory predicts that the age of the earth is about 25 million years. The penetration depth in km for this time is

$$\text{In[12]:= } \sqrt{\text{Dfval} * \text{tageval} * \text{speryr}} / 1000 \text{ km}$$

$$\text{Out[12]= } 30.4967 \text{ km}$$

which easily justifies the flat earth approximation.

We will discuss this value for the age in section 6 below. Our next step will be to view graphically the solution we have just found.

5. A Graphical View of the Solution

We construct here a sequence of graphs of the solution versus x for various times. On each graph, we also show the surface slope for that time (in blue) and the present-day value of the surface slope (in red). We then animate the graph sequence, and stop the animation when the two slopes coincide. The associated time is then the age of the earth. The earlier estimate of the penetration depth suggests that an x -scale running from 0 to 100 km will be adequate.

We start by defining the curves to be plotted. The first is the temperature, evaluated for the parameter values we have chosen.

```
In[13]:= Tval[x_, t_] := T[x, t] /. Thread[{Ti, Ts, Df, α} -> {Tival, Tsval, Dfval, αval}]
```

Next we define the surface slope β_{val} at time t , and the tangent line with that slope, tanline:

```
In[14]:= βval[t_] := β[t] /. Thread[{Ti, Ts, Df, α} -> {Tival, Tsval, Dfval, αval}]
```

```
In[15]:= βval[t]
```

```
Out[15]= 
$$\frac{1.03006 \times 10^6}{\sqrt{t}}$$

```

```
In[16]:= Tval[x, t]
```

```
Out[16]= 
$$2000 \operatorname{Erf}\left[\frac{456.435 x}{\sqrt{t}}\right]$$

```

```
In[17]:= tanline[x_, t_] = Tsval + βval[t] * x
```

```
Out[17]= 
$$\frac{1.03006 \times 10^6 x}{\sqrt{t}}$$

```

The last curve to be defined is the tangent line, nowtanline, based on the present-day slope:

```
In[18]:= nowtanline[x_] = Tsval + αval * x
```


```
Out[18]= 0.037 x
```

We now define graph[t] to be a graph of the solution with the two tangent lines at time t , where t is in years:

```
In[19]:= graph[t_] := Module[{xscale, tscale}, xscale = 1000; tscale = speryr;
  Plot[{Tval[xscale * x, tscale * t], tanline[xscale * x, tscale * t], nowtanline[xscale * x]},
    {x, 0, 100}, PlotRange -> {Tsval, Tival}, AxesLabel -> {"x (km)", "T (°C)"},
    PlotLabel -> Row[{"t = ", PaddedForm[t, {4, 1}], " yr"}],
    PlotStyle -> {{RGBColor[0, 0, 0], Thickness[0.004]},
      {RGBColor[0, 0, 1], Thickness[0.004]}, {RGBColor[1, 0, 0], Thickness[0.004]}}]]
```

Let's try this for $t = 10^6$ years.


```
In[20]:= graph[10.^6]
```



This graph shows that the gradient at 10^6 years is much larger than the present-day gradient.




Now we use Manipulate to construct a sequence of graphs to illustrate the determination of the age of the earth. We run the sequence from 1 million to 40 million years. Using the slider, we can easily show that the predicted theoretical gradient (blue tangent line) coincides with the present observed gradient (red line) when t is approximately 25 million years. We can also play the graphs as a movie to conclude the same thing.




```
In[21]:= Manipulate[graph[i * 10.0^6], {i, 1, 40}]
```




For the printed version of this notebook, we do not have the interactive Manipulate display. We print the few of the graphs at increments of 5 million years.

```
In[22]:= Module[{inc, i}, inc = 5 * 10.^6; Do[Print[graph[i * inc]], {i, 1, 8}]]
```







These graphs again show that the result predicted by this theory for the age of the earth is 25 million years.

6. Discussion of Kelvin's Answer

It is impossible to discuss all of the relevant issues here. For a full discussion, see the book by Burchfield listed below. We give here only a very brief summary.

The answer of 25 million years found by Kelvin was not received favorably by geologists. Both the physical geologists and paleontologists could point to evidence that much more time was needed to produce what they saw in the stratigraphic and fossil records. As one answer to his critics, Kelvin produced a completely independent estimate -- this time for the age of the sun. His result was in close agreement with his estimate of the age of the earth. The solar estimate was based on the idea that the energy supply for the solar radiative flux is gravitational contraction. These two independent and agreeing estimates of the age of two primary members of the solar system formed a strong case for the correctness of his answer.

As we now know, both of Kelvin's answers were wrong, for reasons that he could not have known. In the case of the earth, the energy released in the crust by the decay of radioactive elements is sufficient to enhance significantly the geothermal flux. This larger flux then causes Kelvin's theory to predict an earth too young. In the case of the sun, Kelvin had no way of knowing that the energy source is nuclear fusion, for which there is fuel enough for at least 5×10^9 years.

That Kelvin was wrong does not change at all that what he did was good science. His pioneering of a quantitative approach to such fundamental questions was his lasting contribution in this area.

7. References

A brief treatment of Kelvin's analysis with descriptions of some subsequent work is given in **Conduction of Heat in Solids**, by H.S. Carslaw and J.C. Jaeger, section 2.14, second edition, Oxford Press, 1959. Values of the basic parameters are given there, and the basic mathematical solution is given in section 2.4.

The definitive (and book-length) treatment of Kelvin's work is given in **Lord Kelvin and the Age of the Earth**, J.D. Burchfield, University of Chicago Press, 1990.